Drinfeld modules

Release 10.0

The Sage Development Team

CONTENTS

1	Drinfeld modules	3			
	1.1 Drinfeld modules				
	1.2 Finite Drinfeld modules	16			
2	Morphisms and isogenies	23			
	2.1 Drinfeld module morphisms				
	2.2 Set of morphisms between two Drinfeld modules	26			
3	The module action induced by a Drinfeld module				
	3.1 The module action induced by a Drinfeld module	29			
4	The category of Drinfeld modules				
	4.1 Drinfeld modules over a base	31			
5 Indices and Tables					
Рy	Python Module Index				
In	ndex				

 $Sage Math\ include\ facilities\ to\ manipulate\ Drinfeld\ modules\ and\ their\ morphisms.\ The\ main\ entry\ point\ is\ the\ class\ sage.rings.function_field_modules.drinfeld_modules.DrinfeldModule.$

CONTENTS 1

2 CONTENTS

CHAPTER

ONE

DRINFELD MODULES

1.1 Drinfeld modules

This module provides the class sage.rings.function_field.drinfeld_module.drinfeld_module. DrinfeldModule.

For finite Drinfeld modules and their theory of complex multiplication, see class sage.rings.function_field. drinfeld_module.finite_drinfeld_module.DrinfeldModule.

AUTHORS:

- Antoine Leudière (2022-04)
- Xavier Caruso (2022-06)

class sage.rings.function_field.drinfeld_modules.drinfeld_module.DrinfeldModule(gen,

category)

Bases: Parent, UniqueRepresentation

This class implements Drinfeld $\mathbb{F}_q[T]$ -modules.

Let $\mathbb{F}_q[T]$ be a polynomial ring with coefficients in a finite field \mathbb{F}_q and let K be a field. Fix a ring morphism $\gamma: \mathbb{F}_q[T] \to K$; we say that K is an $\mathbb{F}_q[T]$ -field. Let $K\{\tau\}$ be the ring of Ore polynomials with coefficients in K, whose multiplication is given by the rule $\tau \lambda = \lambda^q \tau$ for any $\lambda \in K$.

A Drinfeld $\mathbb{F}_q[T]$ -module over the base $\mathbb{F}_q[T]$ -field K is an \mathbb{F}_q -algebra morphism $\phi: \mathbb{F}_q[T] \to K\{\tau\}$ such that $\operatorname{Im}(\phi) \not\subset K$ and ϕ agrees with γ on \mathbb{F}_q .

For a in $\mathbb{F}_q[T]$, $\phi(a)$ is denoted ϕ_a .

The Drinfeld $\mathbb{F}_q[T]$ -module ϕ is uniquely determined by the image ϕ_T of T; this serves as input of the class.

Note: See also sage.categories.drinfeld_modules.

The base morphism is the morphism $\gamma: \mathbb{F}_q[T] \to K$. The monic polynomial that generates the kernel of γ is called the $\mathbb{F}_a[T]$ -characteristic, or function-field characteristic, of the base field. We say that $\mathbb{F}_a[T]$ is the function ring of ϕ ; $K\{\tau\}$ is the Ore polynomial ring. Further, the generator is ϕ_T and the constant coefficient is the constant coefficient of ϕ_T .

A Drinfeld module is said to be *finite* if the field K is. Despite an emphasis on this case, the base field can be any extension of \mathbb{F}_q :

```
sage: Fq = GF(25)
sage: A.<T> = Fq[]
sage: K.\langle z \rangle = Fq.extension(6)
```

(continues on next page)

```
sage: phi = DrinfeldModule(A, [z, 4, 1])
sage: phi
Drinfeld module defined by T |--> t^2 + 4*t + z
```

```
sage: Fq = GF(49)
sage: A.<T> = Fq[]
sage: K = Frac(A)
sage: psi = DrinfeldModule(A, [K(T), T+1])
sage: psi
Drinfeld module defined by T |--> (T + 1)*t + T
```

Note: Finite Drinfeld modules are implemented in the class sage.rings.function_field.drinfeld_modules.finite_drinfeld_module.

Classical references on Drinfeld modules include [Gos1998], [Rosen2002], [VS06] and [Gek1991].

Note: Drinfeld modules are defined in a larger setting, in which the polynomial ring $\mathbb{F}_q[T]$ is replaced by a more general function ring: the ring of functions in k that are regular outside ∞ , where k is a function field over \mathbb{F}_q with transcendence degree 1 and ∞ is a fixed place of k. This is out of the scope of this implementation.

INPUT:

- function_ring a univariate polynomial ring whose base field is a finite field
- gen the generator of the Drinfeld module; as a list of coefficients or an Ore polynomial
- name (default: 't') the name of the Ore polynomial ring generator

Construction

A Drinfeld module object is constructed by giving the function ring and the generator:

```
sage: Fq.<z2> = GF(3^2)
sage: A.<T> = Fq[]
sage: K.<z> = Fq.extension(6)
sage: phi = DrinfeldModule(A, [z, 1, 1])
sage: phi
Drinfeld module defined by T |--> t^2 + t + z
```

Note: Note that the definition of the base field is implicit; it is automatically defined as the compositum of all the parents of the coefficients.

The above Drinfeld module is finite: it can also be infinite:

```
sage: L = Frac(A)
sage: psi = DrinfeldModule(A, [L(T), 1, T^3 + T + 1])
sage: psi
Drinfeld module defined by T |--> (T^3 + T + 1)*t^2 + t + T
```

```
sage: phi.is_finite()
True
sage: psi.is_finite()
False
```

In those examples, we used a list of coefficients ([z, 1, 1]) to represent the generator $\phi_T = z + t + t^2$. One can also use regular Ore polynomials:

```
sage: ore_polring = phi.ore_polring()
sage: t = ore_polring.gen()
sage: rho_T = z + t^3
sage: rho = DrinfeldModule(A, rho_T)
sage: rho
Drinfeld module defined by T |--> t^3 + z
sage: rho(T) == rho_T
True
```

Images under the Drinfeld module are computed by calling the object:

```
sage: phi(T) # phi_T, the generator of the Drinfeld module
t^2 + t + z
sage: phi(T^3 + T + 1) # phi_(T^3 + T + 1)
t^6 + (z^11 + z^9 + 2*z^6 + 2*z^4 + 2*z + 1)*t^4 + (2*z^11 + 2*z^10 + z^9 + z^8 + 2*z^7 + 2*z^6 + z^5 + 2*z^3)*t^3 + (2*z^11 + z^10 + z^9 + 2*z^7 + 2*z^6 + z^5 + z^4 + 2*z^3 + 2*z + 2)*t^2 + (2*z^11 + 2*z^8 + 2*z^6 + z^5 + z^4 + 2*z^2)*t + z^3 + 2*z + 1
sage: phi(1) # phi_1
1
```

The category of Drinfeld modules

Drinfeld modules have their own category (see class sage.categories.drinfeld_modules. DrinfeldModules):

```
sage: phi.category()
Category of Drinfeld modules over Finite Field in z of size 3^12 over its base
sage: phi.category() is psi.category()
False
sage: phi.category() is rho.category()
True
```

One can use the category to directly create new objects:

```
sage: cat = phi.category()
sage: cat.object([z, 0, 0, 1])
Drinfeld module defined by T |--> t^3 + z
```

1.1. Drinfeld modules 5

The base field of a Drinfeld module

The base field of the Drinfeld module is retrieved using base():

```
sage: phi.base()
Finite Field in z of size 3^12 over its base
```

The base morphism is retrieved using base_morphism():

```
sage: phi.base_morphism()
Ring morphism:
   From: Univariate Polynomial Ring in T over Finite Field in z2 of size 3^2
   To: Finite Field in z of size 3^12 over its base
   Defn: T |--> z
```

Note that the base field is *not* the field K. Rather, it is a ring extension (see sage.rings.ring_extension. RingExtension) whose underlying ring is K and whose base is the base morphism:

```
sage: phi.base() is K
False
```

Getters

One can retrieve basic properties:

```
sage: phi.base_morphism()
Ring morphism:
  From: Univariate Polynomial Ring in T over Finite Field in z2 of size 3^2
  To: Finite Field in z of size 3^12 over its base
  Defn: T |--> z
```

```
sage: phi.ore_polring() # K{t}
Ore Polynomial Ring in t over Finite Field in z of size 3^12 over its base twisted
→by Frob^2
```

```
sage: phi.function_ring() # Fq[T]
Univariate Polynomial Ring in T over Finite Field in z2 of size 3^2
```

```
sage: phi.gen() # phi_T
t^2 + t + z
sage: phi.gen() == phi(T)
True
```

```
sage: phi.constant_coefficient() # Constant coefficient of phi_T
z
```

```
sage: phi.morphism() # The Drinfeld module as a morphism
Ring morphism:
  From: Univariate Polynomial Ring in T over Finite Field in z2 of size 3^2
  To: Ore Polynomial Ring in t over Finite Field in z of size 3^12 over its base
  →twisted by Frob^2
  Defn: T |--> t^2 + t + z
```

One can compute the rank and height:

```
sage: phi.rank()
2
sage: phi.height()
1
```

As well as the j-invariant if the rank is two:

```
sage: phi.j_invariant() # j-invariant
1
```

A Drinfeld $\mathbb{F}_q[T]$ -module can be seen as an Ore polynomial with positive degree and constant coefficient $\gamma(T)$, where γ is the base morphism. This analogy is the motivation for the following methods:

```
sage: phi.coefficients()
[z, 1, 1]
```

```
sage: phi.coefficient(1)
1
```

Morphisms and isogenies

A morphism of Drinfeld modules $\phi \to \psi$ is an Ore polynomial $f \in K\{\tau\}$ such that $f\phi_a = \psi_a f$ for every a in the function ring. In our case, this is equivalent to $f\phi_T = \psi_T f$. An isogeny is a nonzero morphism.

Use the in syntax to test if an Ore polynomial defines a morphism:

```
sage: phi(T) in Hom(phi, phi)
True
sage: t^6 in Hom(phi, phi)
True
sage: t^5 + 2*t^3 + 1 in Hom(phi, phi)
False
sage: 1 in Hom(phi, rho)
False
sage: 1 in Hom(phi, phi)
True
sage: 0 in Hom(phi, rho)
True
```

To create a SageMath object representing the morphism, call the homset (hom):

```
sage: hom = Hom(phi, phi)
sage: frobenius_endomorphism = hom(t^6)
sage: identity_morphism = hom(1)
sage: zero_morphism = hom(0)
sage: frobenius_endomorphism
Endomorphism of Drinfeld module defined by T |--> t^2 + t + z
    Defn: t^6
sage: identity_morphism
Identity morphism of Drinfeld module defined by T |--> t^2 + t + z
sage: zero_morphism
```

(continues on next page)

1.1. Drinfeld modules 7

```
Endomorphism of Drinfeld module defined by T |--> t^2 + t + z
Defn: 0
```

The underlying Ore polynomial is retrieved with the method ore_polynomial():

```
sage: frobenius_endomorphism.ore_polynomial()
t^6
sage: identity_morphism.ore_polynomial()
1
```

One checks if a morphism is an isogeny, endomorphism or isomorphism:

```
sage: frobenius_endomorphism.is_isogeny()
True
sage: identity_morphism.is_isogeny()
True
sage: zero_morphism.is_isogeny()
False
sage: frobenius_endomorphism.is_isomorphism()
False
sage: identity_morphism.is_isomorphism()
True
sage: zero_morphism.is_isomorphism()
False
```

The Vélu formula

Let P be a nonzero Ore polynomial. We can decide if P defines an isogeny with a given domain and, if it does, find the codomain:

If the input does not define an isogeny, an exception is raised:

```
sage: phi.velu(0)
Traceback (most recent call last):
...
ValueError: the input does not define an isogeny
sage: phi.velu(t)
Traceback (most recent call last):
...
ValueError: the input does not define an isogeny
```

The action of a Drinfeld module

The $\mathbb{F}_q[T]$ -Drinfeld module ϕ induces a special left $\mathbb{F}_q[T]$ -module structure on any field extension L/K. Let $x \in L$ and a be in the function ring; the action is defined as $(a,x) \mapsto \phi_a(x)$. The method action() returns a sage.rings.function_field.drinfeld_modules.action.Action object representing the Drinfeld module action.

Note: In this implementation, L is K:

```
sage: action = phi.action()
sage: action
Action on Finite Field in z of size 3^12 over its base induced by Drinfeld module_
    defined by T |--> t^2 + t + z
```

The action on elements is computed by calling the action object:

```
sage: P = T + 1
sage: a = z
sage: action(P, a)
...
z^9 + 2*z^8 + 2*z^7 + 2*z^6 + 2*z^3 + z^2
sage: action(0, K.random_element())
0
sage: action(A.random_element(), 0)
```

Warning: The class DrinfeldModuleAction may be replaced later on. See issues #34833 and #34834.

action()

Return the action object (sage.rings.function_field.drinfeld_modules.action.Action) that represents the module action, on the base codomain, that is induced by the Drinfeld module.

OUTPUT: a Drinfeld module action object

EXAMPLES:

The action on elements is computed as follows:

1.1. Drinfeld modules 9

```
sage: P = T^2 + T + 1
sage: a = z12 + 1
sage: action(P, a)
3*z12^11 + 2*z12^10 + 3*z12^9 + 3*z12^7 + 4*z12^5 + z12^4 + z12^3 + 2*z12 + 1
sage: action(0, a)
0
sage: action(P, 0)
0
```

coefficient(n)

Return the n-th coefficient of the generator.

INPUT:

• n – a nonnegative integer

OUTPUT: an element in the base codomain

EXAMPLES:

```
sage: Fq = GF(25)
sage: A.<T> = Fq[]
sage: K.<z12> = Fq.extension(6)
sage: p_root = 2*z12^11 + 2*z12^10 + z12^9 + 3*z12^8 + z12^7 + 2*z12^5 + 2*z12^6
\rightarrow4 + 3*z12^3 + z12^2 + 2*z12
sage: phi = DrinfeldModule(A, [p_root, z12^3, z12^5])
sage: phi.coefficient(0)
2*z12^11 + 2*z12^10 + z12^9 + 3*z12^8 + z12^7 + 2*z12^5 + 2*z12^4 + 3*z12^3 +
\hookrightarrowz12^2 + 2*z12
sage: phi.coefficient(0) == p_root
True
sage: phi.coefficient(1)
z12^3
sage: phi.coefficient(2)
z12^5
sage: phi.coefficient(5)
Traceback (most recent call last):
ValueError: input must be >= 0 and <= rank
```

coefficients(sparse=True)

Return the coefficients of the generator, as a list.

If the flag sparse is True (default), only return the nonzero coefficients; otherwise, return all of them.

INPUT:

• sparse – a boolean

EXAMPLES:

(continues on next page)

Careful, the method only returns the nonzero coefficients, unless otherwise specified:

gen()

Return the generator of the Drinfeld module.

EXAMPLES:

height()

Return the height of the Drinfeld module if the function field characteristic is a prime ideal; raise ValueError otherwise.

The height of a Drinfeld module is defined when the function field characteristic is a prime ideal. In our case, this ideal is even generated by a monic polynomial $\mathfrak p$ in the function field. Write $\phi_{\mathfrak p}=a_s\tau^s+\cdots+\tau^{r*\deg(\mathfrak p)}$. The height of the Drinfeld module is the well-defined positive integer $h=\frac{s}{\deg(\mathfrak p)}$.

Note: See [Gos1998], Definition 4.5.8 for the general definition.

A rank two Drinfeld module is supersingular if and only if its height equals its rank.

EXAMPLES:

```
sage: Fq = GF(25)
sage: A.<T> = Fq[]
sage: K.<z12> = Fq.extension(6)
```

(continues on next page)

1.1. Drinfeld modules

```
sage: B.<Y> = Fq[]
sage: L = Frac(B)
sage: phi = DrinfeldModule(A, [L(2), L(1)])
sage: phi.height()
Traceback (most recent call last):
...
NotImplementedError: height not implemented in this case
```

```
sage: Fq = GF(343)
sage: A.<T> = Fq[]
sage: K.<z6> = Fq.extension(2)
sage: phi = DrinfeldModule(A, [1, 0, z6])
sage: phi.height()
2
sage: phi.is_supersingular()
True
```

is_finite()

Return True if this Drinfeld module is finite, False otherwise.

EXAMPLES:

j_invariant()

Return the j-invariant of the Drinfeld module if the rank is two; raise a NotImplementedError otherwise.

Assume the rank is two. Write the generator $\phi_T = \omega + g\tau + \Delta \tau^2$. The j-invariant is defined by $\frac{g^{q+1}}{\Delta}$, q being the order of the base field of the function ring. In our case, this field is always finite.

OUTPUT: an element in the base codomain

EXAMPLES:

The rank must be two:

```
sage: sigma = DrinfeldModule(A, [p_root, 1, 0])
sage: sigma.j_invariant()
Traceback (most recent call last):
...
NotImplementedError: rank must be 2
```

morphism()

Return the morphism object that defines the Drinfeld module.

OUTPUT: a ring morphism from the function ring to the Ore polynomial ring

EXAMPLES:

```
sage: Fq = GF(25)
sage: A.<T> = Fq[]
sage: K.<z12> = Fq.extension(6)
sage: p_{root} = 2*z12^{11} + 2*z12^{10} + z12^{9} + 3*z12^{8} + z12^{7} + 2*z12^{5} + 2*z12^{6}
 4 + 3*z12^3 + z12^2 + 2*z12
sage: phi = DrinfeldModule(A, [p_root, z12^3, z12^5])
sage: phi.morphism()
Ring morphism:
       From: Univariate Polynomial Ring in T over Finite Field in z2 of size 5^2
                              Ore Polynomial Ring in t over Finite Field in z12 of size 5^12 over its.
 ⇒base twisted by Frob^2
    Defn: T \mid --> z12^5*t^2 + z12^3*t + 2*z12^11 + 2*z12^10 + z12^9 + 3*z12^8 + 2*z12^10 +
 \Rightarrowz12^7 + 2*z12^5 + 2*z12^4 + 3*z12^3 + z12^2 + 2*z12
sage: from sage.rings.morphism import RingHomomorphism
sage: isinstance(phi.morphism(), RingHomomorphism)
True
```

Actually, the DrinfeldModule method __call__() simply class the __call__ method of this morphism:

```
sage: phi.morphism()(T) == phi(T)
True
sage: a = A.random_element()
```

(continues on next page)

1.1. Drinfeld modules 13

```
sage: phi.morphism()(a) == phi(a)
True
```

And many methods of the Drinfeld module have a counterpart in the morphism object:

rank()

Return the rank of the Drinfeld module.

In our case, the rank is the degree of the generator.

OUTPUT: an integer

EXAMPLES:

velu(isog)

Return a new Drinfeld module such that input is an isogeny to this module with domain self; if no such isogeny exists, raise an exception.

INPUT:

• isog – the Ore polynomial that defines the isogeny

OUTPUT: a Drinfeld module

ALGORITHM:

The input defines an isogeny if only if:

- 1. The degree of the characteristic divides the height of the input. (The height of an Ore polynomial $P(\tau)$ is the maximum n such that τ^n right-divides $P(\tau)$.)
- 2. The input right-divides the generator, which can be tested with Euclidean division.

We test if the input is an isogeny, and, if it is, we return the quotient of the Euclidean division.

Height and Euclidean division of Ore polynomials are implemented as methods of class sage. rings.polynomial.ore_polynomial_element.OrePolynomial.

Another possible algorithm is to recursively solve a system, see arXiv 2203.06970, Eq. 1.1.

EXAMPLES:

```
sage: Fq = GF(25)
sage: A.<T> = Fq[]
sage: K.<z12> = Fq.extension(6)
sage: p_{root} = 2*z12^{11} + 2*z12^{10} + z12^{9} + 3*z12^{8} + z12^{7} + 2*z12^{5} + 2*z12^{6}
 \rightarrow4 + 3*z12^3 + z12^2 + 2*z12
sage: phi = DrinfeldModule(A, [p_root, z12<sup>3</sup>, z12<sup>5</sup>])
sage: t = phi.ore_polring().gen()
sage: isog = t + 2*z12^11 + 4*z12^9 + 2*z12^8 + 2*z12^6 + 3*z12^5 + z12^4 + 
\rightarrow 2*z12^3 + 4*z12^2 + 4*z12 + 4
sage: psi = phi.velu(isog)
sage: psi
Drinfeld module defined by T \mid -- \rangle (z12^11 + 3*z12^10 + z12^9 + z12^7 + z12^5 +__
 4*z12^4 + 4*z12^3 + z12^2 + 1*12^2 + 1*12^3 + 212^4 + 4*z12^10 + 2*z12^8 + z12^6 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 + 2*12^8 +
 \rightarrow3*z12^5 + z12^4 + 2*z12^3 + z12^2 + z12 + 4)*t + 2*z12^11 + 2*z12^10 + z12^9_
 \hookrightarrow + 3*z12^8 + z12^7 + 2*z12^5 + 2*z12^4 + 3*z12^3 + z12^2 + 2*z12
sage: isog in Hom(phi, psi)
True
```

This method works for endomorphisms as well:

```
sage: phi.velu(phi(T)) is phi
True
sage: phi.velu(t^6) is phi
True
```

The following inputs do not define isogenies, and the method returns None:

```
sage: phi.velu(0)
Traceback (most recent call last):
...
ValueError: the input does not define an isogeny
sage: phi.velu(t)
Traceback (most recent call last):
...
ValueError: the input does not define an isogeny
sage: phi.velu(t^3 + t + 2)
Traceback (most recent call last):
...
ValueError: the input does not define an isogeny
```

1.1. Drinfeld modules 15

1.2 Finite Drinfeld modules

This module provides the class sage.rings.function_fields.drinfeld_module.finite_drinfeld_module. FiniteDrinfeldModule, which inherits sage.rings.function_fields.drinfeld_module. drinfeld module.DrinfeldModule.

AUTHORS:

• Antoine Leudière (2022-04)

 $\textbf{class} \texttt{ sage.rings.function_field.drinfeld_modules.finite_drinfeld_module.} \textbf{FiniteDrinfeldModule} (\textit{gen}, \textit{permitted finite f$

category)

Bases: DrinfeldModule

This class implements finite Drinfeld $\mathbb{F}_q[T]$ -modules.

A *finite Drinfeld module* is a Drinfeld module whose base field is finite. In this case, the function field characteristic is a prime ideal.

For general definitions and help on Drinfeld modules, see class sage.rings.function_fields.drinfeld_module.drinfeld_module.DrinfeldModule.

Construction:

The user does not ever need to directly call FiniteDrinfeldModule — the metaclass DrinfeldModule is responsible for instantiating DrinfeldModule or FiniteDrinfeldModule depending on the input:

```
sage: Fq = GF(343)
sage: A.<T> = Fq[]
sage: K.<z6> = Fq.extension(2)
sage: phi = DrinfeldModule(A, [z6, 0, 5])
sage: phi
Drinfeld module defined by T |--> 5*t^2 + z6
```

```
sage: isinstance(phi, DrinfeldModule)
True
sage: from sage.rings.function_field.drinfeld_modules.finite_drinfeld_module import

→FiniteDrinfeldModule
sage: isinstance(phi, FiniteDrinfeldModule)
True
```

The user should never use FiniteDrinfeldModule to test if a Drinfeld module is finite, but rather the is_finite method:

```
sage: phi.is_finite()
True
```

Complex multiplication of rank two finite Drinfeld modules

We can handle some aspects of the theory of complex multiplication of finite Drinfeld modules. Apart from the method frobenius_endomorphism, we only handle rank two Drinfeld modules.

First of all, it is easy to create the Frobenius endomorphism:

```
sage: frobenius_endomorphism = phi.frobenius_endomorphism()
sage: frobenius_endomorphism
Endomorphism of Drinfeld module defined by T |--> 5*t^2 + z6
Defn: t^2
```

Its characteristic polynomial can be computed:

```
sage: chi = phi.frobenius_charpoly()
sage: chi
X^2 + (T + 2*z3^2 + 2*z3 + 1)*X + 2*T^2 + (z3^2 + z3 + 4)*T + 2*z3
sage: frob_pol = frobenius_endomorphism.ore_polynomial()
sage: chi(frob_pol, phi(T))
0
```

as well as its trace and norm:

```
sage: phi.frobenius_trace()
6*T + 5*z3^2 + 5*z3 + 6
sage: phi.frobenius_trace() == -chi[1]
True
sage: phi.frobenius_norm()
2*T^2 + (z3^2 + z3 + 4)*T + 2*z3
```

We can decide if a Drinfeld module is ordinary or supersingular:

```
sage: phi.is_ordinary()
True
sage: phi.is_supersingular()
False
```

Inverting the Drinfeld module

The morphism that defines a Drinfeld module is injective (see [Gos1998], cor. 4.5.2). If the Drinfeld module is finite, one can retrieve preimages:

```
sage: a = A.random_element()
sage: phi.invert(phi(a)) == a
True
```

frobenius_charpoly(var='X')

Return the characteristic polynomial of the Frobenius endomorphism if the rank is two. Raise a NotImplementedError otherwise.

Let \mathbb{F}_q be the base field of the function ring. The *characteristic polynomial `chi` of the Frobenius endo-morphism* is defined in [Gek1991]. An important feature of this polynomial is that it is a monic univariate polynomial with coefficients in the function ring. As in our case the function ring is a univariate polynomial ring, it is customary to see the characteristic polynomial of the Frobenius endomorphism as a bivariate polynomial.

Let $\chi = X^2 - A(T)X + B(T)$ be the characteristic polynomial of the Frobenius endomorphism, and let t^n be the Ore polynomial that defines the Frobenius endomorphism of ϕ ; by definition, n is the degree over of the base field over \mathbb{F}_q . We have $\chi(t^n)(\phi(T)) = t^{2n} - \phi_A t^n + \phi_B = 0$, with $\deg(A) \leq \frac{n}{2}$ and $\deg(B) = n$.

Note that the *Frobenius trace* is defined as A(T) and the *Frobenius norm* is defined as B(T).

INPUT:

• var (default: 'X') – the name of the second variable

EXAMPLES:

```
sage: Fq = GF(343)
sage: A.<T> = Fq[]
sage: K.<z6> = Fq.extension(2)
sage: phi = DrinfeldModule(A, [1, 0, z6])
sage: chi = phi.frobenius_charpoly()
sage: chi
X^2 + ((3*z3^2 + z3 + 4)*T + 4*z3^2 + 6*z3 + 3)*X + (5*z3^2 + 2*z3)*T^2 + (4*z3^2 + 3*z3)*T + 5*z3^2 + 2*z3
```

```
sage: frob_pol = phi.frobenius_endomorphism().ore_polynomial()
sage: chi(frob_pol, phi(T))
0
```

```
sage: trace = phi.frobenius_trace()
sage: trace
(4*z3^2 + 6*z3 + 3)*T + 3*z3^2 + z3 + 4
sage: norm = phi.frobenius_norm()
sage: norm
(5*z3^2 + 2*z3)*T^2 + (4*z3^2 + 3*z3)*T + 5*z3^2 + 2*z3
```

```
sage: n = 2 # Degree of the base field over Fq
sage: trace.degree() <= n/2
True
sage: norm.degree() == n
True</pre>
```

ALGORITHM:

We compute the Frobenius norm, and with it the Frobenius trace. This gives the Frobenius characteristic polynomial. See [MS2019], Section 4.

See docstrings of methods *frobenius_norm()* and *frobenius_trace()* for further details on the computation of the norm and of the trace.

frobenius_endomorphism()

Return the Frobenius endomorphism of the Drinfeld module as a morphism object.

Let q be the order of the base field of the function ring. The *Frobenius endomorphism* is defined as the endomorphism whose defining Ore polynomial is t^q .

EXAMPLES:

```
sage: Fq = GF(343)
sage: A.<T> = Fq[]
```

(continues on next page)

```
sage: K.<z6> = Fq.extension(2)
sage: phi = DrinfeldModule(A, [1, 0, z6])
sage: phi.frobenius_endomorphism()
Endomorphism of Drinfeld module defined by T |--> z6*t^2 + 1
    Defn: t^2
```

frobenius_norm()

Return Frobenius norm of the Drinfeld module, if the rank is two, raise a NotImplementedError otherwise.

Let $\mathbb{F}_q[T]$ be the function ring, write $\chi=X^2-A(T)X+B(T)\in\mathbb{F}_q[T][X]$ for the characteristic polynomial of the Frobenius endomorphism. The *Frobenius norm* is defined as the polynomial $B(T)\in\mathbb{F}_q[T]$.

Let n be the degree of the base field over \mathbb{F}_q Then the Frobenius norm has degree n.

EXAMPLES:

```
sage: Fq = GF(343)
sage: A.<T> = Fq[]
sage: K.<z6> = Fq.extension(2)
sage: phi = DrinfeldModule(A, [1, 0, z6])
sage: B = phi.frobenius_norm()
sage: B
(5*z3^2 + 2*z3)*T^2 + (4*z3^2 + 3*z3)*T + 5*z3^2 + 2*z3
```

```
sage: n = 2 # Degree of the base field over Fq
sage: B.degree() == n
True
```

```
sage: B == phi.frobenius_charpoly()[0]
True
```

ALGORITHM:

The Frobenius norm is computed using the formula, by Gekeler, given in [MS2019], Section 4, Proposition 3.

frobenius_trace()

Return Frobenius norm of the Drinfeld module, if the rank is two; raise a NotImplementedError otherwise.

Let $\mathbb{F}_q[T]$ be the function ring, write $\chi=T^2-A(X)T+B(X)\in\mathbb{F}_q[T][X]$ for the characteristic polynomial of the Frobenius endomorphism. The *Frobenius trace* is defined as the polynomial $A(T)\in\mathbb{F}_q[T]$.

Let n be the degree over \mathbb{F}_q of the base codomain. Then the Frobenius trace has degree at most $\frac{n}{2}$.

EXAMPLES:

```
sage: Fq = GF(343)
sage: A.<T> = Fq[]
sage: K.<z6> = Fq.extension(2)
sage: phi = DrinfeldModule(A, [1, 0, z6])
sage: A = phi.frobenius_trace()
sage: A
(4*z3^2 + 6*z3 + 3)*T + 3*z3^2 + z3 + 4
```

```
sage: n = 2 # Degree over Fq of the base codomain
sage: A.degree() <= n/2
True</pre>
```

```
sage: A == -phi.frobenius_charpoly()[1]
True
```

ALGORITHM:

Let A(T) denote the Frobenius trace and B(T) denote the Frobenius norm. We begin by computing B(T), see docstring of method $frobenius_norm()$ for details. The characteristic polynomial of the Frobenius yields $t^{2n} - \phi_A t^n + \phi_B = 0$, where t^n is the Frobenius endomorphism. As ϕ_B is now known, we can compute $\phi_A = (t^{2n} + \phi_B)/t^n$. We get A(T) by inverting this quantity, using the method sage.rings.function_fields.drinfeld_module.drinfeld_module. DrinfeldModule.invert(), see its docstring for details.

invert(ore_pol)

Return the preimage of the input under the Drinfeld module, if it exists.

INPUT:

• ore_pol – the Ore polynomial whose preimage we want to compute

EXAMPLES:

When the input is not in the image of the Drinfeld module, an exception is raised:

```
sage: t = phi.ore_polring().gen()
sage: phi.invert(t + 1)
Traceback (most recent call last):
...
ValueError: input must be in the image of the Drinfeld module

sage: phi.invert(t^3 + t^2 + 1)
Traceback (most recent call last):
...
ValueError: input must be in the image of the Drinfeld module
```

ALGORITHM:

The algorithm relies on the inversion of a linear algebra system. See [MS2019], 3.2.5 for details.

is_ordinary()

Return True whether the Drinfeld module is ordinary; raise a NotImplementedError if the rank is not two.

A rank two finite Drinfeld module is *ordinary* if and only if the function ring-characteristic does not divide the Frobenius trace. A *supersingular* rank two finite Drinfeld module is a Drinfeld module that is not ordinary.

A rank two Drinfeld module is *ordinary* if and only if it is not supersingular; see *is_supersingular()*.

EXAMPLES:

```
sage: Fq = GF(343)
sage: A.<T> = Fq[]
sage: K.<z6> = Fq.extension(2)
sage: phi = DrinfeldModule(A, [1, 0, z6])
sage: phi.is_ordinary()
False
sage: phi_p = phi(phi.characteristic())
sage: phi_p # Purely inseparable
z6*t^2
```

ALGORITHM:

Compute the Frobenius trace and test if the $\mathbb{F}_q[T]$ characteristic divides it.

We could also test if the image of the $\mathbb{F}_q[T]$ -characteristic under the Drinfeld module is purely inseparable; see [Gek1991], Proposition 4.1.

is_supersingular()

Return True whether the Drinfeld module is supersingular; raise a NotImplementedError if the rank is not two.

A rank two finite Drinfeld module is *supersingular* if and only if the function ring-characteristic divides the Frobenius trace. An *ordinary* rank two finite Drinfeld module is a Drinfeld module that is not supersingular.

EXAMPLES:

```
sage: Fq = GF(343)
sage: A.<T> = Fq[]
sage: K.<z6> = Fq.extension(2)
sage: phi = DrinfeldModule(A, [1, 0, z6])
sage: phi.is_supersingular()
True
sage: phi(phi.characteristic()) # Purely inseparable
z6*t^2
```

ALGORITHM:

Compute the Frobenius trace and test if the function ring-characteristic divides it.

We could also test if the image of the function ring-characteristic under the Drinfeld module is purely inseparable; see [Gek1991], Proposition 4.1.

CHAPTER

TWO

MORPHISMS AND ISOGENIES

2.1 Drinfeld module morphisms

This module provides the class sage.rings.function_fields.drinfeld_module.morphism. DrinfeldModuleMorphism.

AUTHORS: - Antoine Leudière (2022-04)

Bases: Morphism, UniqueRepresentation

This class represents Drinfeld $\mathbb{F}_q[T]$ -module morphisms.

Let ϕ and ψ be two Drinfeld $\mathbb{F}_q[T]$ -modules over a field K. A morphism of Drinfeld modules $\phi \to \psi$ is an Ore polynomial $f \in K\{\tau\}$ such that $f\phi_a = \psi_a f$ for every $a \in \mathbb{F}_q[T]$. In our case, this is equivalent to $f\phi_T = \psi_T f$. An isogeny is a nonzero morphism.

To create a morphism object, the user should never explicitly instantiate *DrinfeldModuleMorphism*, but rather call the parent homset with the defining Ore polynomial:

The given Ore polynomial must indeed define a morphism:

```
sage: morphism = Hom(phi, psi)(1)
Traceback (most recent call last):
...
ValueError: Ore polynomial does not define a morphism
```

One can get basic data on the morphism:

```
sage: morphism.domain()
Drinfeld module defined by T |--> (z^2 + z)*t^2 + (z^2 + z)*t + z
sage: morphism.domain() is phi
True

sage: morphism.codomain()
Drinfeld module defined by T |--> (z^5 + z^2 + z + 1)*t^2 + (z^4 + z + 1)*t + z
sage: morphism.codomain() is psi
True
```

```
sage: morphism.ore_polynomial()
t + z^5 + z^3 + z + 1
sage: morphism.ore_polynomial() is ore_pol
True
```

One can check various properties:

```
sage: morphism.is_zero()
False
sage: morphism.is_isogeny()
True
sage: morphism.is_endomorphism()
False
sage: morphism.is_isomorphism()
False
```

is_identity()

Return True whether the morphism is the identity morphism.

EXAMPLES:

```
sage: Fq = GF(2)
sage: A.<T> = Fq[]
sage: K.<z6> = Fq.extension(6)
sage: phi = DrinfeldModule(A, [z6, 1, 1])
sage: morphism = End(phi)(1)
sage: morphism.is_identity()
True
```

```
sage: psi = DrinfeldModule(A, [z6, z6^4 + z6^2 + 1, 1])
sage: t = phi.ore_polring().gen()
sage: morphism = Hom(phi, psi)(t + z6^5 + z6^2 + 1)
sage: morphism.is_identity()
False
```

is_isogeny()

Return True whether the morphism is an isogeny.

EXAMPLES:

```
sage: Fq = GF(2)
sage: A.<T> = Fq[]
sage: K.<z6> = Fq.extension(6)
```

(continues on next page)

```
sage: phi = DrinfeldModule(A, [z6, 1, 1])
sage: psi = DrinfeldModule(A, [z6, z6^4 + z6^2 + 1, 1])
sage: t = phi.ore_polring().gen()
sage: morphism = Hom(phi, psi)(t + z6^5 + z6^2 + 1)
sage: morphism.is_isogeny()
True
```

```
sage: zero_morphism = End(phi)(0)
sage: zero_morphism.is_isogeny()
False
```

```
sage: identity_morphism = End(phi)(1)
sage: identity_morphism.is_isogeny()
True
```

```
sage: frobenius_endomorphism = phi.frobenius_endomorphism()
sage: frobenius_endomorphism.is_isogeny()
True
```

is_isomorphism()

Return True whether the morphism is an isomorphism.

EXAMPLES:

```
sage: Fq = GF(2)
sage: A.<T> = Fq[]
sage: K.<z6> = Fq.extension(6)
sage: phi = DrinfeldModule(A, [z6, 1, 1])
sage: psi = DrinfeldModule(A, [z6, z6^4 + z6^2 + 1, 1])
sage: t = phi.ore_polring().gen()
sage: morphism = Hom(phi, psi)(t + z6^5 + z6^2 + 1)
sage: morphism.is_isomorphism()
False
```

```
sage: zero_morphism = End(phi)(0)
sage: zero_morphism.is_isomorphism()
False
```

```
sage: identity_morphism = End(phi)(1)
sage: identity_morphism.is_isomorphism()
True
```

```
sage: frobenius_endomorphism = phi.frobenius_endomorphism()
sage: frobenius_endomorphism.is_isomorphism()
False
```

is_zero()

Return True whether the morphism is the zero morphism.

EXAMPLES:

```
sage: Fq = GF(2)
sage: A.<T> = Fq[]
sage: K.<z6> = Fq.extension(6)
sage: phi = DrinfeldModule(A, [z6, 1, 1])
sage: psi = DrinfeldModule(A, [z6, z6^4 + z6^2 + 1, 1])
sage: t = phi.ore_polring().gen()
sage: morphism = Hom(phi, psi)(t + z6^5 + z6^2 + 1)
sage: morphism.is_zero()
False
```

```
sage: zero_morphism = End(phi)(0)
sage: zero_morphism.is_zero()
True
```

ore_polynomial()

Return the Ore polynomial that defines the morphism.

EXAMPLES:

```
sage: Fq = GF(2)
sage: A.<T> = Fq[]
sage: K.<z6> = Fq.extension(6)
sage: phi = DrinfeldModule(A, [z6, 1, 1])
sage: psi = DrinfeldModule(A, [z6, z6^4 + z6^2 + 1, 1])
sage: t = phi.ore_polring().gen()
sage: morphism = Hom(phi, psi)(t + z6^5 + z6^2 + 1)
sage: ore_pol = morphism.ore_polynomial()
sage: ore_pol
t + z6^5 + z6^2 + 1
```

```
sage: ore_pol * phi(T) == psi(T) * ore_pol
True
```

2.2 Set of morphisms between two Drinfeld modules

This module provides the class sage.rings.function_field.drinfeld_module.homset. DrinfeldModuleHomset.

AUTHORS:

• Antoine Leudière (2022-04)

Bases: Homset

This class implements the set of morphisms between two Drinfeld $\mathbb{F}_q[T]$ -modules.

INPUT:

- X the domain
- Y the codomain

EXAMPLES:

```
sage: Fq = GF(27)
sage: A.<T> = Fq[]
sage: K.<z6> = Fq.extension(2)
sage: phi = DrinfeldModule(A, [z6, z6, 2])
sage: psi = DrinfeldModule(A, [z6, 2*z6^5 + 2*z6^4 + 2*z6 + 1, 2])
sage: H = Hom(phi, psi)
sage: H
Set of Drinfeld module morphisms from (gen) 2*t^2 + z6*t + z6 to (gen) 2*t^2 + z6*t + z
```

```
sage: from sage.rings.function_field.drinfeld_modules.homset import

→DrinfeldModuleHomset
sage: isinstance(H, DrinfeldModuleHomset)
True
```

There is a simpler syntax for endomorphisms sets:

The domain and codomain must have the same Drinfeld modules category:

```
sage: rho = DrinfeldModule(A, [Frac(A)(T), 1])
sage: Hom(phi, rho)
Traceback (most recent call last):
...
ValueError: Drinfeld modules must be in the same category
```

```
sage: sigma = DrinfeldModule(A, [1, z6, 2])
sage: Hom(phi, sigma)
Traceback (most recent call last):
...
ValueError: Drinfeld modules must be in the same category
```

One can create morphism objects by calling the homset:

```
sage: identity_morphism = E(1)
sage: identity_morphism
Identity morphism of Drinfeld module defined by T |--> 2*t^2 + z6*t + z6
```

```
sage: t = phi.ore_polring().gen()
sage: frobenius_endomorphism = E(t^6)
sage: frobenius_endomorphism
Endomorphism of Drinfeld module defined by T |--> 2*t^2 + z6*t + z6
Defn: t^6
```

And one can test if an Ore polynomial defines a morphism using the in syntax:

```
sage: 1 in H
False
sage: t^6 in H
False
sage: t + 1 in H
True
sage: 1 in E
True
sage: t^6 in E
True
sage: t + 1 in E
False
```

This also works if the candidate is a morphism object:

```
sage: isogeny in H
True
sage: E(0) in E
True
sage: identity_morphism in H
False
sage: frobenius_endomorphism in H
False
```

Element

alias of DrinfeldModuleMorphism

CHAPTER

THREE

THE MODULE ACTION INDUCED BY A DRINFELD MODULE

3.1 The module action induced by a Drinfeld module

This module provides the class sage.rings.function_field.drinfeld_module.action. DrinfeldModuleAction.

AUTHORS:

• Antoine Leudière (2022-04)

 $\textbf{class} \texttt{ sage.rings.function_field.drinfeld_modules.action.} \textbf{DrinfeldModuleAction} (\textit{drinfeld_module})$

Bases: Action

This class implements the module action induced by a Drinfeld $\mathbb{F}_q[T]$ -module.

Let ϕ be a Drinfeld $\mathbb{F}_q[T]$ -module over a field K and let L/K be a field extension. Let $x \in L$ and let a be a function ring element; the action is defined as $(a,x) \mapsto \phi_a(x)$.

Note: In this implementation, L is K.

Note: The user should never explicitly instantiate the class DrinfeldModuleAction.

Warning: This class may be replaced later on. See issues #34833 and #34834.

INPUT: the Drinfeld module

EXAMPLES:

The action on elements is computed as follows:

```
sage: P = T + 1
sage: a = z
sage: action(P, a)
...
4*z + 2
sage: action(0, K.random_element())
0
sage: action(A.random_element(), 0)
```

Finally, given a Drinfeld module action, it is easy to recover the corresponding Drinfeld module:

```
sage: action.drinfeld_module() is phi
True
```

drinfeld_module()

Return the Drinfeld module defining the action.

OUTPUT: a Drinfeld module

EXAMPLES:

```
sage: Fq.<z2> = GF(11)
sage: A.<T> = Fq[]
sage: K.<z> = Fq.extension(2)
sage: phi = DrinfeldModule(A, [z, 0, 0, 1])
sage: action = phi.action()
sage: action.drinfeld_module() is phi
True
```

CHAPTER

FOUR

THE CATEGORY OF DRINFELD MODULES

4.1 Drinfeld modules over a base

This module provides the class sage.category.drinfeld_modules.DrinfeldModules.

AUTHORS:

- Antoine Leudière (2022-04)
- Xavier Caruso (2022-06)

class sage.categories.drinfeld_modules.DrinfeldModules(base_field, name='t')

```
Bases: Category_over_base_ring
```

This class implements the category of Drinfeld $\mathbb{F}_q[T]$ -modules on a given base field.

Let $\mathbb{F}_q[T]$ be a polynomial ring with coefficients in a finite field \mathbb{F}_q and let K be a field. Fix a ring morphism $\gamma: \mathbb{F}_q[T] \to K$; we say that K is an $\mathbb{F}_q[T]$ '*-field*.Let' $K\{\tau\}$ be the ring of Ore polynomials with coefficients in K, whose multiplication is given by the rule $\tau\lambda = \lambda^q\tau$ for any $\lambda \in K$.

The extension $K/\mathbb{F}_q[T]$ (represented as an instance of the class sage.rings.ring_extension. RingExtension) is the *base field* of the category; its defining morphism γ is called the *base morphism*.

The monic polynomial that generates the kernel of γ is called the $\mathbb{F}_q[T]$ -characteristic, or function-field characteristic, of the base field. We say that $\mathbb{F}_q[T]$ is the function ring of the category; $K\{\tau\}$ is the Ore polynomial ring. The constant coefficient of the category is the image of T under the base morphism.

Construction

Generally, Drinfeld modules objects are created before their category, and the category is retrieved as an attribute of the Drinfeld module:

```
sage: Fq = GF(11)
sage: A.<T> = Fq[]
sage: K.<z> = Fq.extension(4)
sage: p_root = z^3 + 7*z^2 + 6*z + 10
sage: phi = DrinfeldModule(A, [p_root, 0, 0, 1])
sage: C = phi.category()
sage: C
Category of Drinfeld modules over Finite Field in z of size 11^4 over its base
```

The output tells the user that the category is only defined by its base.

Properties of the category

The base field is retrieved using the method base().

sage: C.base() Finite Field in z of size 11⁴ over its base

Equivalently, one can use *base_morphism()* to retrieve the base morphism:

```
sage: C.base_morphism()
Ring morphism:
  From: Univariate Polynomial Ring in T over Finite Field of size 11
  To: Finite Field in z of size 11^4 over its base
  Defn: T |--> z^3 + 7*z^2 + 6*z + 10
```

The so-called constant coefficient — which is the same for all Drinfeld modules in the category — is simply the image of T by the base morphism:

```
sage: C.constant_coefficient()
z^3 + 7*z^2 + 6*z + 10
sage: C.base_morphism()(T) == C.constant_coefficient()
True
```

Similarly, the function ring-characteristic of the category is either 0 or the unique monic polynomial in $\mathbb{F}_q[T]$ that generates the kernel of the base:

```
sage: C.characteristic()
T^2 + 7*T + 2
sage: C.base_morphism()(C.characteristic())
0
```

The base field, base morphism, function ring and Ore polynomial ring are the same for the category and its objects:

Creating Drinfeld module objects from the category

Calling *object()* with an Ore polynomial creates a Drinfeld module object in the category whose generator is the input:

```
sage: psi = C.object([p_root, 1])
sage: psi
Drinfeld module defined by T |--> t + z^3 + 7*z^2 + 6*z + 10
sage: psi.category() is C
True
```

Of course, the constant coefficient of the input must be the same as the category:

```
sage: C.object([z, 1])
Traceback (most recent call last):
...
ValueError: constant coefficient must equal that of the category
```

It is also possible to create a random object in the category. The input is the desired rank:

Endsets()

Return the category of endsets.

EXAMPLES:

```
sage: Fq = GF(11)
sage: A.<T> = Fq[]
sage: K.<z> = Fq.extension(4)
sage: p_root = z^3 + 7*z^2 + 6*z + 10
sage: phi = DrinfeldModule(A, [p_root, 0, 0, 1])
sage: C = phi.category()
sage: from sage.categories.homsets import Homsets
sage: C.Endsets() is Homsets().Endsets()
True
```

Homsets()

Return the category of homsets.

EXAMPLES:

```
sage: Fq = GF(11)
sage: A.<T> = Fq[]
sage: K.<z> = Fq.extension(4)
sage: p_root = z^3 + 7*z^2 + 6*z + 10
sage: phi = DrinfeldModule(A, [p_root, 0, 0, 1])
```

(continues on next page)

(continued from previous page)

```
sage: C = phi.category()
sage: from sage.categories.homsets import Homsets
sage: C.Homsets() is Homsets()
True
```

class ParentMethods

Bases: object

base()

Return the base field of this Drinfeld module, viewed as an algebra over the function ring.

This is an instance of the class sage.rings.ring_extension.RingExtension.

EXAMPLES:

The base can be infinite:

```
sage: sigma = DrinfeldModule(A, [Frac(A).gen(), 1])
sage: sigma.base()
Fraction Field of Univariate Polynomial Ring in T over Finite Field in z2

→ of size 5^2 over its base
```

base_morphism()

Return the base morphism of this Drinfeld module.

EXAMPLES:

The base field can be infinite:

```
sage: sigma = DrinfeldModule(A, [Frac(A).gen(), 1])
sage: sigma.base_morphism()
```

(continues on next page)

(continued from previous page)

```
Ring morphism:

From: Univariate Polynomial Ring in T over Finite Field in z2 of size 5^2
To: Fraction Field of Univariate Polynomial Ring in T over Finite Field

in z2 of size 5^2 over its base

Defn: T |--> T
```

base_over_constants_field()

Return the base field, seen as an extension over the constants field \mathbb{F}_q .

This is an instance of the class ${\tt sage.rings.ring_extension.RingExtension.}$

EXAMPLES:

characteristic()

Return the function ring-characteristic.

EXAMPLES:

constant_coefficient()

Return the constant coefficient of the generator of this Drinfeld module.

OUTPUT: an element in the base field

Let $\mathbb{F}_q[T]$ be the function ring, and let γ be the base of the Drinfeld module. The constant coefficient is $\gamma(T)$:

```
sage: C = phi.category()
sage: base = C.base()
sage: base(T) == phi.constant_coefficient()
True
```

Naturally, two Drinfeld modules in the same category have the same constant coefficient:

Reciprocally, it is impossible to create two Drinfeld modules in this category if they do not share the same constant coefficient:

```
sage: rho = C.object(phi.constant_coefficient() + 1 + t^3)
Traceback (most recent call last):
...
ValueError: constant coefficient must equal that of the category
```

function_ring()

Return the function ring of this Drinfeld module.

EXAMPLES:

ore_polring()

Return the Ore polynomial ring of this Drinfeld module.

The Ore polynomial ring can also be retrieved from the category of the Drinfeld module:

```
sage: S is phi.category().ore_polring()
True
```

The generator of the Drinfeld module is in the Ore polynomial ring:

```
sage: phi(T) in S
True
```

base_morphism()

Return the base morphism of the category.

EXAMPLES:

```
sage: Fq = GF(11)
sage: A.<T> = Fq[]
sage: K.<z> = Fq.extension(4)
sage: p_root = z^3 + 7*z^2 + 6*z + 10
sage: phi = DrinfeldModule(A, [p_root, 0, 0, 1])
sage: C = phi.category()
sage: C.base_morphism()
Ring morphism:
   From: Univariate Polynomial Ring in T over Finite Field of size 11
   To: Finite Field in z of size 11^4 over its base
   Defn: T |--> z^3 + 7*z^2 + 6*z + 10
sage: C.constant_coefficient() == C.base_morphism()(T)
True
```

base_over_constants_field()

Return the base field, seen as an extension over the constants field \mathbb{F}_q .

```
sage: Fq = GF(11)
sage: A.<T> = Fq[]
sage: K.<z> = Fq.extension(4)
sage: p_root = z^3 + 7*z^2 + 6*z + 10
sage: phi = DrinfeldModule(A, [p_root, 0, 0, 1])
sage: C = phi.category()
sage: C.base_over_constants_field()
Field in z with defining polynomial x^4 + 8*x^2 + 10*x + 2 over its base
```

characteristic()

Return the function ring-characteristic.

EXAMPLES:

```
sage: Fq = GF(11)
sage: A.<T> = Fq[]
sage: K.<z> = Fq.extension(4)
sage: p_root = z^3 + 7*z^2 + 6*z + 10
sage: phi = DrinfeldModule(A, [p_root, 0, 0, 1])
sage: C = phi.category()
sage: C.characteristic()
T^2 + 7*T + 2
```

```
sage: psi = DrinfeldModule(A, [Frac(A).gen(), 1])
sage: C = psi.category()
sage: C.characteristic()
0
```

constant_coefficient()

Return the constant coefficient of the category.

EXAMPLES:

```
sage: Fq = GF(11)
sage: A.<T> = Fq[]
sage: K.<z> = Fq.extension(4)
sage: p_root = z^3 + 7*z^2 + 6*z + 10
sage: phi = DrinfeldModule(A, [p_root, 0, 0, 1])
sage: C = phi.category()
sage: C.constant_coefficient()
z^3 + 7*z^2 + 6*z + 10
sage: C.constant_coefficient() == C.base()(T)
True
```

function_ring()

Return the function ring of the category.

EXAMPLES:

```
sage: Fq = GF(11)
sage: A.<T> = Fq[]
sage: K.<z> = Fq.extension(4)
sage: p_root = z^3 + 7*z^2 + 6*z + 10
sage: phi = DrinfeldModule(A, [p_root, 0, 0, 1])
sage: C = phi.category()
sage: C.function_ring()
Univariate Polynomial Ring in T over Finite Field of size 11
sage: C.function_ring() is A
True
```

object(gen)

Return a Drinfeld module object in the category whose generator is the input.

INPUT:

• gen – the generator of the Drinfeld module, given as an Ore polynomial or a list of coefficients

EXAMPLES:

```
sage: Fq = GF(11)
sage: A.<T> = Fq[]
sage: K.<z> = Fq.extension(4)
sage: p_root = z^3 + 7*z^2 + 6*z + 10
sage: psi = DrinfeldModule(A, [p_root, 1])
sage: C = psi.category()

sage: phi = C.object([p_root, 0, 1])
sage: phi
Drinfeld module defined by T |--> t^2 + z^3 + 7*z^2 + 6*z + 10
sage: t = phi.ore_polring().gen()
sage: C.object(t^2 + z^3 + 7*z^2 + 6*z + 10) is phi
True
```

ore_polring()

Return the Ore polynomial ring of the category

EXAMPLES:

```
sage: Fq = GF(11)
sage: A.<T> = Fq[]
sage: K.<z> = Fq.extension(4)
sage: p_root = z^3 + 7*z^2 + 6*z + 10
sage: phi = DrinfeldModule(A, [p_root, 0, 0, 1])
sage: C = phi.category()
sage: C.ore_polring()
Ore Polynomial Ring in t over Finite Field in z of size 11^4 over its base_
    →twisted by Frob
```

random_object(rank)

Return a random Drinfeld module in the category with given rank.

INPUT:

• rank – an integer, the rank of the Drinfeld module

EXAMPLES:

```
sage: Fq = GF(11)
sage: A.<T> = Fq[]
sage: K.<z> = Fq.extension(4)
sage: p_root = z^3 + 7*z^2 + 6*z + 10
sage: phi = DrinfeldModule(A, [p_root, 0, 0, 1])
sage: C = phi.category()

sage: psi = C.random_object(3) # random
Drinfeld module defined by T |--> (6*z^3 + 4*z^2 + 10*z + 9)*t^3 + (4*z^3 + 8*z^4 + 2 + 8*z)*t^2 + (10*z^3 + 3*z^2 + 6*z)*t + z^3 + 7*z^2 + 6*z + 10
sage: psi.rank() == 3
True
```

super_categories()

```
sage: Fq = GF(11)
sage: A.<T> = Fq[]
sage: K.<z> = Fq.extension(4)
sage: p_root = z^3 + 7*z^2 + 6*z + 10
sage: phi = DrinfeldModule(A, [p_root, 0, 0, 1])
sage: C = phi.category()
sage: C.super_categories()
[Category of objects]
```

CHAPTER

FIVE

INDICES AND TABLES

- Index
- Module Index
- Search Page

PYTHON MODULE INDEX

```
C
sage.categories.drinfeld_modules, 31

r
sage.rings.function_field.drinfeld_modules.action,
29
sage.rings.function_field.drinfeld_modules.drinfeld_module,
3
sage.rings.function_field.drinfeld_modules.finite_drinfeld_module,
16
sage.rings.function_field.drinfeld_modules.homset,
26
sage.rings.function_field.drinfeld_modules.morphism,
```

44 Python Module Index

INDEX

A		dModuleAction	(class	in
<pre>action() (sage.rings.function_field.drinfeld_modules.drin</pre>	feld_modu	de:BrinfeaaMbatilen- 29	field.drinfeld_modules.a	ıction),
memou),)	Drinfel	dModuleHomset	(class	in
В			field.drinfeld_modules.h	iomset),
base() (sage.categories.drinfeld_modules.DrinfeldModule method), 34	s ParentM Drinfel	<i>léthods</i> dModuleMorphism	(class	in
method), 34 base_morphism() (sage.categories.drinfeld_modules.Drin method), 37			field.drinfeld_modules.n	norphism),
base_morphism() (sage.categories.drinfeld_modules.Drin method), 34	Drinfel <i>nfeldModu</i>	dModules les.ParentMethods sage.categories.drin	(class feld_modules), 31	in
base_over_constants_field()	Drinfel	dModules.ParentM	ethods (class	in
$(sage. categories. drinfeld_modules. Drinfeld Modules and the property of th$	les	sage.categories.drin	feld_modules), 34	
method), 37	Е			
<pre>base_over_constants_field()</pre>	les.Parent	(Aggerings.function_	field.drinfeld_modules.l	homset.DrinfeldModu
method), 35	Endooto	attribute), 28	winfold modules Drinfol	ldModulos
С	Eliusets	method), 33	rinfeld_modules.Drinfel	amounes
<pre>characteristic() (sage.categories.drinfeld_modules.Dr method), 37</pre>	i nf eldMod	ules		
characteristic() (sage.categories.drinfeld_modules.Dr method), 35	ihjeldMod	rinfoldModule utes:ParentMethods sage.rings.function_	(class field.drinfeld_modules.fi	in înite_drinfeld_module
coefficient() (sage.rings.function_field.drinfeld_module method), 10	frobeni	us_charpoly()		
coefficients() (sage.rings.function_field.drinfeld_module method), 10	ıles.drinfel	asassains Diunfeiam method), 17	.ftg.ld_edrinfeld_modules.	finite_drinfeld_modul
constant_coefficient()	frobeni	us_endomorphism()	
(sage.categories.drinfeld_modules.DrinfeldModu method), 38	les	(sage.rings.function_method), 18	_field.drinfeld_modules.	finite_drinfeld_modul
constant_coefficient()	frobeni	us_norm()(sage.rin	gs.function_field.drinfel	d_modules.finite_drir
(sage.categories.drinfeld_modules.DrinfeldModu	les Parent	method), 19		•
method), 35	frobeni	us_trace() (sage.ri method), 19	ngs.function_field.drinfe	eld_modules.finite_dr
D	function	n ring()(sage cate	gories.drinfeld_modules	s DrinfoldModules
		method)38	gories.arinjeta_mountes	.Drinjeiamoanies
drinfeld_module() (sage.rings.function_field.drinfeld_n method), 30	iodules.ac function	tion:DrinfeldModules n_ring() (sage.cate	Action gories.drinfeld_modules	s.DrinfeldModules.Pa
DrinfeldModule (class in		method), 36		
sage.rings.function_field.drinfeld_modules.drinfe	l G module	?),		
-	gen() (sa	age.rings.function_fie	ld.drinfeld_modules.dri	nfeld_module.Drinfel

method), 11

```
Η
                                                                                                   R
height() (sage.rings.function_field.drinfeld_modules.drinfeldhodules.DrinfeldModules.DrinfeldModules
                                                                                                                   method), 39
                method), 11
Homsets() (sage.categories.drinfeld modules.DrinfeldModnbank() (sage.rings.function field.drinfeld modules.drinfeld module.Drinfeldmodnbank()
                method), 33
                                                                                                                   method), 14
                                                                                                   S
invert() (sage.rings.function_field.drinfeld_modules.finite_dringelddtegdnlesEindriDrinfelddModules
                                                                                                           module, 31
                method), 20
is_finite() (sage.rings.function_field.drinfeld_modules.dsiageldrimgbul&uDvanfebdMbilald.drinfeld_modules.action
                                                                                                           module, 29
                method), 12
is_identity()(sage.rings.function_field.drinfeld_modules.avgerphingsDrfunfedtMondufaModphisinnfeld_modules.drinfeld_modules.
                method), 24
                                                                                                           module, 3
is_isogeny() (sage.rings.function_field.drinfeld_modules.sageplasing?rinfahdMiodulefiblichrinfeld_modules.finite_drinfeld
                                                                                                           module, 16
                method), 24
is_isomorphism() (sage.rings.function_field_drinfeld_modalge.nninphssnfullvintfedaMdrinhifeld_modules.homset
                                                                                                           module, 26
                method), 25
is_ordinary() (sage.rings.function field.drinfeld modules.figieteridrinfeldfuncothion_FiniteNovidfeldNewblul_modules.morphism
                method), 20
                                                                                                           module, 23
is_supersingular() (sage.rings.function_field.drinfeld_nswebels_fixttegdrinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles.drinfeld_nswebeles
                                                                                                                   method), 39
is_zero() (sage.rings.function_field.drinfeld_modules.morphism.DrinfeldModuleMorphism
                method), 25
                                                                                                   velu() (sage.rings.function_field.drinfeld_modules.drinfeld_module.Drinfe
J
                                                                                                                   method), 14
\verb|j_invariant()| (sage.rings.function\_field.drinfeld\_modules.drinfeld\_module.DrinfeldModule)|
                method), 12
M
module
        sage.categories.drinfeld_modules, 31
        sage.rings.function_field.drinfeld_modules.action,
                29
        sage.rings.function_field.drinfeld_modules.drinfeld_module,
        sage.rings.function_field.drinfeld_modules.finite_drinfeld_module,
        sage.rings.function_field.drinfeld_modules.homset,
        sage.rings.function_field.drinfeld_modules.morphism,
morphism() (sage.rings.function_field.drinfeld_modules.drinfeld_module.DrinfeldModule
                method), 13
object() (sage.categories.drinfeld_modules.DrinfeldModules
                method), 38
ore_polring() (sage.categories.drinfeld_modules.DrinfeldModules
                method), 39
ore_polring() (sage.categories.drinfeld_modules.DrinfeldModules.ParentMethods
                method), 36
ore_polynomial() (sage.rings.function field.drinfeld modules.morphism.DrinfeldModuleMorphism
                method), 26
```

46 Index